
&  3.89 FKLABEL()
! รูปแบบ
FKLABEL(<เลขระบุฟงกชัน>) --> ชื่อปุมฟงกชัน
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สงชื่อฟงกชันเปนตัวอักษร
: ตัวอยางที่ 3.103
        INKEY(0)
        ? FKLABEL(1)            // F1
        ? FKLABEL(20)           // F20
        ? FKLABEL(35)           // F35
        ? FKMAX()               //         40

&  3.90 FKMAX()
! รูปแบบ
FKMAX() --> เลขระบุฟงกชันสูงสุดที่เปนไปได
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สงคาสูงสุดของปุมฟงกชันที่เปนไปได ไมใหเกินขอกํ าหนด
: ตัวอยางที่ 3.104
        INKEY(0)     ;      ? FKMAX()               // 40 เลขฟงกชันที่ใชไดสูงสุด

&  3.91 FLOCK()
! รูปแบบ
FLOCK() --> ผลการจองแฟม
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ไมอนุญาติใหมีการใชฐานขอมูลรวมกัน มีประโยชนมากกับระบบเครือขาย
: ตัวอยางที่ 3.105
        USE FILEA SHARED        // เปดแฟมในพื้นที ่1 และไมจองไว
        USE FILEA SHARED NEW // เปดแฟมในพื้นที ่2 และไมจองไว
        ? FIELD1                // 101
        ? FLOCK()               // .T. จองแฟมไมใหใครแกไข ในพื้นที ่2
        SELE 1 ;    ? FIELD1        // 101 น ําขอมูลมาแสดงได
        //  DELETE                 แตไมสามารถลบได เพราะแฟมถูก LOCK
        ? FLOCK()               // .F.

: ตัวอยางที่ 3.106
        USE FILEA
        USE FILEA EXCLUSIVE NEW // เปดพรอมระบุไมใหมีการแบงใชขอมูล
        SELE 1
        ? FIELD1                // 101 น ําขอมูลมาแสดงได
        ? FLOCK()               // .T.
        DELETE                  // ลบขอมูลไดเพราะจองไวแกไข
        RECALL ALL
        SELE 2
        // FIELD1               น ําฟลดมาใชไมไดเพราะเปดแบบ EXCLUSIVE
        ? FLOCK()               // .F.
: ตัวอยางที่ 3.107
        USE FILEA EXCLUSIVE     // เปดพรอมระบุไมใหมีการแบงใชขอมูล
        USE FILEA SHARED NEW
        SELE 1
          ? FIELD1                // 101 น ําขอมูลมาแสดงได
          ? FLOCK()               // .T.
          DELETE                  // ลบขอมูลไดเพราะจองไวแกไขแลว
          RECALL ALL
        SELE 2
          // FIELD1    น ําฟลดมาใชไมไดเพราะพื้นที ่1 เปดแบบ EXCLUSIVE
          ? FLOCK()               // .F.
        SELE 1
          CLOSE                   // ท ําใหแฟมในพื้นที่ 2 ปดเชนกัน
        SELE 2
          USE FILEA SHARED
          ? FLOCK()               // .T.
          ? FIELD1                // 101
          UNLOCK                  // UNLOCK ตองใชในพื้นที่ที ่LOCK
        SELE 1  ;     USE FILEA SHARED
          ? FLOCK()               // .T.
        SELE 2  ;      ? FLOCK()  // .F.

&  3.92 FOPEN()
! รูปแบบ
FOPEN(<ชือ่แฟม>, [<แบบในการเปดแฟม>]) --> ตัวแปรระดับแฟม
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สั่งเปดแฟมแบบ BINARY FILE

เปดแฟมได 3 ลักษณะดังนี้
0 เปดแฟมแบบ อานไดอยางเดียว
1 เปดแฟมแบบ เขียนใหเขียนลงไปไดอยางเดียว
2 เปดแฟมแบบ อนุญาติใหทั้งอานและเขียนลงไปในแฟม

: ตัวอยางที่ 3.108
        F = FOPEN("TEST.TXT",2)
        ? FREADSTR(F,1000)      // น ํา 1000 ไบตแรกมาแสดง หรือเทาที่มี
        FSEEK(F,0,0)            // เลื่อนตัวชี้ไปที่ตนแฟม
        ? FWRITE(F,"TEST",4)    // 4  จ ํานวนไบตที่เขียนลงไปในแฟม

&  3.93 FOUND()
! รูปแบบ
FOUND() --> ผลการคนหา
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สงผลของการ SEARCH เชน LOCATE หรือ SEEK
: ตัวอยางที่ 3.109
        USE FILEA
        LOCATE FOR FIELD2 > 900
        DO WHILE FOUND()
           ? FIELD1,FIELD2,FIELD3
           SKIP
           LOCATE REST FOR FIELD2 > 900
        ENDDO
: ตัวอยางที่ 3.110
        USE FILEA
        CRITERIA = "FIELD2 > 900"
        LOCATE FOR &CRITERIA
        I = 1

        DO WHILE FOUND()
           ? I,FIELD1,FIELD2,FIELD3
           SKIP
           LOCATE REST FOR &CRITERIA
           I++
        ENDDO
: ตัวอยางที่ 3.111
        USE FILEA
        LOCATE FOR FIELD1 = '102'
        IF FOUND() ; ? FIELD1,FIELD2,FIELD3 ; ENDIF
        GO TOP   ;        LOCATE FOR FIELD2 < 1000
        IF FOUND() ; ? FIELD1,FIELD2,FIELD3 ; ENDIF
        GO TOP   ;        LOCATE FOR FIELD1 = '102'
        ? IF(FOUND(),(FIELD1+STR(FIELD2)+STR(FIELD3)),'NOT FOUND')
        ? IIF(.NOT. FOUND(),,FIELD2+FIELD3) // ถาไมพบจะไมแสดงอะไร

&  3.94 FREAD()
! รูปแบบ
FREAD(<ตวัแปรระดับแฟม>, @<ตัวแปรรับอักขระ>, <จํ านวนอักขระ>) --> จํ านวนอักขระ
P หนาที่
ฟงกชันนี้จะท ําหนาที่ อานแฟมแบบ BINARY FILE แลวสงคาใหตัวแปร
: ตัวอยางที่ 3.112
        X = 10                          // ตวัแปรบอกขนาดการอานแตละครั้ง
        GETDATA = SPACE(X)              // ตวัแปรที่รับคาจากการอาน
        F = FOPEN("C:\AUTOEXEC.BAT",0)  // เปดแฟมแบบอานอยางเดียว
        DO WHILE .T.
           IF FREAD(F,@GETDATA,X) <> X
              ? "ERROR READING"         // ตวัอกัษรที่เศษจาก 10
              EXIT                      // จะไมถูกพิมพออกมา
           ELSE
              ?? GETDATA
           ENDIF
        ENDDO
        FCLOSE(F)

&  3.95 FREADSTR()
! รูปแบบ
FREADSTR(<ตวัแปรระดับแฟม>, <จํ านวนอักขระ>) --> ขอความที่อานได
P หนาที่
ฟงกชันนี้จะท ําหนาที่ อานแฟมแบบ BINARY FILE หรือ สั่งใหอานโดยไมสงคา
: ตัวอยางที่ 3.113
        F = FOPEN("C:\AUTOEXEC.BAT",0)  // เปดแฟมแบบอานอยางเดียว
        X = " "                         // ก ําหนดคาเริ่มตนให X
        DO WHILE LEN(X) <> 0
           X = FREADSTR(F,40)           // ขนาดของ X ขยายเอง
           ?? X
        ENDDO
        FCLOSE(F)
: ตัวอยางที่ 3.114
        USE FILEA
        H  = HEADER()
        RC = RECCOUNT()
        RZ = RECSIZE()
        CLOSE ALL
        F = FOPEN("FILEA.DBF",0)        // เปดแฟมแบบอานอยางเดียว
        FREADSTR(F,H)                   // อานหัวของแฟมขอมูล
        FOR I = 1 TO RC
            ? FREADSTR(F,RZ)            // น ําขอมูลในเรคอรดมาพิมพ
        NEXT
        FCLOSE(F)

&  3.96 FRENAME()
! รูปแบบ
FRENAME(<ชือ่แฟมเดิม>, <ชื่อแฟมใหม>) --> ผลการเปลี่ยนชื่อ
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สั่งเปลี่ยนชื่อแฟม
การเปลีย่นชื่อที่ถูกตองจะใหผลเปน 0 แตถาผิดพลาดจะใหผลเปน -1
จากกรณตีาง ๆ เชน แผนเต็ม ชื่อแฟมใหมมีอยูแลว หรือไมพบชื่อแฟมเกาที่จะเปลี่ยน

: ตัวอยางที่ 3.115
        ? FRENAME("X.BAK","X.BKK")        // ถาเปลี่ยนไดจะพิมพเลข 0
        IF FRENAME("X.BAK","X.BKK") = -1  // ผลของบรรทัดนี้มักเปน -1
           ? "ERROR OPERATION ON RENAME"  // เพราะบรรทัดแรกเปลี่ยนชื่อแลว
        ELSE
           ? "O.K."                       // มีโอกาสเปนไปไดนอยที่สุด
        ENDIF
        FRENAME("X.BKK","X.BAK")          // ไมแสดงผลใด ๆ แตจะประมวลผลทันที
        IF FRENAME("X.BAK","X.BKK") = 0 // ประมวลผลพรอมสงคามาตรวจสอบ
           ? "O.K."                       // มโีอกาสเปนไปไดสูง ถามีแฟมชื่อ
        ENDIF                             // X.BAK อยูตั้งแตแรก

&  3.97 FSEEK()
! รูปแบบ
FSEEK(<ตวัแปรระดบัแฟม>, <จํ านวนตํ าแหนง>, [<แบบของการยาย>])
      --> ทีอ่ยูของตํ าแหนงปจจุบัน
P หนาที่
ฟงกชนันี้จะท ําหนาที่ เซตตํ าแหนงของตัวชี้ในแฟมแบบ BINARY FILE

แบบของการยายมี 3 แบบ
0   การยายจากตนแฟมลงไป
1   การยายจากต ําแหนงปจจุบันลงไป
2   การยายจากทายแฟมขึ้นมา

: ตัวอยางที่ 3.116
        F = FOPEN("TEST.TXT")
        ? FSEEK(F,0,0)          //   0 ไปที่ตนแฟม
        ? FSEEK(F,0,1)          //   0 จากตํ าแหนงปจจุบันไปอีก 0 ไบต
        ? FSEEK(F,0,2)          //  15 ไปที่ทายแฟม
        ? FSEEK(F,0,1)          //  15 จากต ําแหนงปจจุบันไปอีก 0 ไบต
        ? FSEEK(F,0,0)          //   0 เปนการเซตคาของตัวชี้ใหม
        ? FSEEK(F,2,0)          //   2 จากตนแฟมไปอีก 2 ไบต
        ? FSEEK(F,5,1)          //   7 จากไบตที่ 2 ไปอีก 5 ไบต

&  3.98 FWRITE()
! รูปแบบ
FWRITE(<ตวัแปรระดบัแฟม>, <ตัวอักษรที่ใสลงแฟม>, [<จํ านวนตัวอักษร>])
       --> จ ํานวนตัวอักษรที่เขียนลงแฟม
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สั่งเขียนลงไปในแฟมแบบ BINARY FILE
: ตัวอยางที่ 3.117
        F = FCREATE("TEST.TXT",0)       // สรางแฟมแบบอานเขียนได
        FWRITE(F,'A')                   // แฟมจะมีขนาด 1 ไบต
        FCLOSE(F)
: ตัวอยางที่ 3.118
        F = FCREATE("TEST.TXT",0)       // สรางแฟมแบบอานเขียนได
        FWRITE(F,'A',10)                // เพิ่มขอมูลในแฟมอีก 10 ไบต
        FWRITE(F,'A',10)                // เพิ่มขอมูลในแฟมอีก 10 ไบต
        FCLOSE(F)                       // แฟมมีขนาด 20 ไบต
        // แตสวนที่ไมไดระบุจะก ําหนดไมไดในแตละ 10 ไบตนั้น
: ตัวอยางที่ 3.119
        CLS
        F = FCREATE("TEST.TXT",0)       // สรางแฟมแบบอานเขียนได
        ? 'พมิพค ําวา END เมื่อตองการเลิกจัดเก็บ'
        DO WHILE .T.
            ACCEPT TO X                 // ขอมูลที่ไดจะไมมีการตัดบรรทัด
            IF X = 'END' ; EXIT ; ENDIF // เพราะไมเก็บรหัสตัดบรรทัดให
            FWRITE(F,X)                 // LASTKEY ใชกับ ACCEPT ไมได
        ENDDO                           // เพราะไมสามารถกด ESC แลวเลิกงาน
        FCLOSE(F)
: ตัวอยางที่ 3.120
        #DEFINE CRLF()  (CHR(13)+CHR(10))
        CLS
        F = FCREATE("TEST.TXT",0)       // สรางแฟมแบบอานเขียนได
        LINE = 2
        @ 1,5 SAY 'เลกิรบัขอมูลใหกด ENTER มาบรรทัดที่วาง แลวกด ESC'
        DO WHILE .T.

            X = SPACE(50)
            @ LINE,5 GET X
            READ
            LINE++
            IF LASTKEY() = 27 ; EXIT ; ENDIF
            FWRITE(F,RTRIM(X)+CRLF())
        ENDDO
        FCLOSE(F)

&  3.99 GETENV()
! รูปแบบ
GETENV(<ชือ่ตวัแปรสภาวะที่เคยเซตไว>) --> คาที่เคยเซตไวใหตัวแปรสภาวะ
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ รับคาที่เคยเซตไวในระบบปฏิบัติการ เพื่อควบคุมสภาพแวดลอม
: ตัวอยางที่ 3.121
        ? GETENV("PATH")        // C:\;C:\CLIPPER5;C:\WINDOWS
        ? GETENV("OBJ")         // C:\CLIPPER5\OBJ
        ? GETENV("LIB")         // C:\CLIPPER5\LIB
        DIR

&  3.100 HARDCR()
! รูปแบบ
HARDCR(<ชดุตวัอกัขระที่ม ีSOFT CR.>) --> ชุดตัวอักขระที่เปน HARD CR.
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แทนที่คา SOFT CARRIAGE RETURN ดวย HARD CARRIAGE RETURN
นัน่คือแทนที่รหัสแอสก ี141 ดวย 13
: ตัวอยางที่ 3.122
        TXT = MEMOREAD("C:\LETTER.TXT")
        TXT = HARDCR(TXT)
        SET CURSOR OFF
        MEMOEDIT(TXT,10,15,20,60,.F.)  // อานขอความมาดู แตไมแกไข
        SET CURSOR ON

&  3.101 HEADER()
! รูปแบบ
HEADER() --> ขนาดของสวนหัวในแฟมขอมูล
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สงคาความยาวของหัวแฟม ส ําหรับแฟมที่นามสกุล DBF
: ตัวอยางที่ 3.123
        USE FILEA
        ? HEADER()              // 130
        ? RECCOUNT()            //   4
        ? RECSIZE()             //  31
        #DEFINE DBFSIZE()  (HEADER() + 1 + (RECCOUNT()*RECSIZE()))
        USE FILEA
        ? DBFSIZE()             // 255

&  3.102 IF()
! รูปแบบ
[I]IF(<เงือ่นไข>, <นิพจนเมื่อเปนจริง>, <นิพจนเมื่อเปนเท็จ>) --> ผลลัพธ
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สงผลการตรวจสอบเงื่อนไข
: ตัวอยางที่ 3.124
        A = 5
        ?  IF(A = 5,"YONOK","LAMPANG")
        ? IIF(A = 5,"YONOK","LAMPANG")
        @ ROW() , IF(A=5,10,20)  SAY "WOW1"
        @ IF(ROW() = 20,1,ROW()) , COL()          SAY "WOW2"
        @ IIF(ROW() = 20,1,ROW()), IIF(A=5,30,40) SAY "WOW3"

&  3.103 INDEXKEY()
! รูปแบบ
INDEXKEY(<ล ําดบัชื่อแฟมดรรชน>ี) --> ชื่อแฟมดรรชนี
P หนาที่
ฟงกชันนี้จะท ําหนาที่ บอกชื่อแฟม INDEX ซึ่งระบุล ําดับของแฟมได

: ตัวอยางที่ 3.125

        USE FILEA INDEX FILEI1,FILEI2,FILEI3
        ? INDEXKEY(0)           // FILEI1
        ? INDEXKEY(1)           // FILEI1
        ? INDEXKEY(2)           // FILEI2
        SET ORDER TO 2
        ? INDEXKEY(0)           // FILEI2   0 หมายถึงแฟม INDEX ปจจุบัน
        ? INDEXKEY(INDEXORD())  // FILEI2

&  3.104 INDEXORD()
! รูปแบบ
INDEXORD() --> ล ําดบัชื่อแฟมดรรชนีที่ก ําลังใชงานอยู
P หนาที่
ฟงกชนันี้จะท ําหนาที่ บอกล ําดับของแฟมแบบ INDEX ที่กํ าลังเปดใชงาน
: ตัวอยางที่ 3.126
        USE FILEA INDEX FILEI1,FILEI2,FILEI3,FILEI4
        ? INDEXORD()                    // 1
        IND := INDEXORD()
        SET ORDER TO 2
        ? INDEXORD()                    // 2
        SET ORDER TO 4
        IF INDEXORD() = 4
           SET ORDER TO IND
        ELSE
           SET ORDER TO 3
        ENDIF

&  3.105 INKEY()
! รูปแบบ
INKEY([<จ ํานวนเวลาเปนวินาท>ี]) --> คาของแปนพิมพที่ถูกกด
P หนาที่
ฟงกชันนี้จะท ําหนาที่ รอคาจาก KEYBOARD

: ตัวอยางที่ 3.127
        INKEY(0)                // WAIT UNTIL KEYPRESSED
        INKEY(1)                // WAIT 1 SECOND
        INKEY(2)                // WAIT 2 SECONDS
        ? "PLEASE KEY TEXT UNTIL ESC"
        CH = SPACE(100)
        X  = INKEY(0)
        ? CHR(X)
        DO WHILE .NOT. LASTKEY() = 27
           X = INKEY(0)
           CH = CH + CHR(X)
           ?? CHR(X)
        ENDDO
        ? CH

&  3.106 INT()
! รูปแบบ
INT(<จ ํานวนเลข>) --> จํ านวนเต็ม
P หนาที่
ฟงกชนันี้จะท ําหนาที่ แปลงจํ านวนเลขเปนจํ านวนเต็ม
: ตัวอยางที่ 3.128
        ? INT(100.25)           //       100
        ? INT(25)               //        25
        ? INT(-25)              //       -25
        ? INT(25.999)           //        25
        ? INT(22 / 7)           //         3

&  3.107 ISALPHA()
! รูปแบบ
ISALPHA(<ขอความ>) --> ผลการตรวจสอบ
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ตรวจสอบวาอักษรตัวแรกเปนตัวอักษรหรือไม

: ตัวอยางที่ 3.129
        ? ISALPHA("ABCD")       // .T.  IT WILL CHECK THE FIRST COL.
        ? ISALPHA("AABB")       // .T.
        ? ISALPHA("1ABB")       // .F.
        ? ISALPHA("1234")       // .F.
        ? ISALPHA("12.4")       // .F.
        ? ISALPHA("A.A.")       // .T.

&  3.108 ISCOLOR()
! รูปแบบ
ISCOLOR() | ISCOLOUR() --> ผลการตรวจจอสี
P หนาที่
ฟงกชันนี้จะท ําหนาที่ ตรวจสอบสีของจอภาพ
: ตัวอยางที่ 3.130
        IF ISCOLOR()
           SETCOLOR("GR+/N,BG+/B")
        ELSE
           SETCOLOR("W/N,N/W")
        ENDIF

&  3.109 ISDIGIT()
! รูปแบบ
ISDIGIT(<ขอความ>) --> ผลการตรวจสอบ
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ตรวจสอบวาอักษรตัวแรกเปนตัวเลขหรือไม
: ตัวอยางที่ 3.131
        ? ISALPHA("ABCD")       // .F.  IT WILL CHECK THE FIRST COL.
        ? ISALPHA("AABB")       // .F.
        ? ISALPHA("1ABB")       // .T.
        ? ISALPHA("1234")       // .T.
        ? ISALPHA("12.4")       // .T.
        ? ISALPHA("A.A.")       // .F.

&  3.110 ISLOWER()
! รูปแบบ
ISLOWER(<ขอความ>) --> ผลการตรวจสอบ
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ตรวจสอบวาอักษรตัวแรกเปนตัวเล็กหรือไม
: ตัวอยางที่ 3.132
        ? ISALPHA("abCD")       // .T.  IT WILL CHECK THE FIRST COL.
        ? ISALPHA("aaBB")       // .T.
        ? ISALPHA("1abb")       // .F.
        ? ISALPHA("1234")       // .F.
        ? ISALPHA("12.4")       // .F.
        ? ISALPHA("A.a.")       // .F.

&  3.111 ISPRINTER()
! รูปแบบ
ISPRINTER() --> ผลการตรวจสอบ
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ตรวจสอบความพรอมของเครื่องพิมพกอนพิมพ
: ตัวอยางที่ 3.133
        IF ISPRINTER()
           SET DEVICE TO PRINTER
           @ PROW(),PCOL() SAY "TEST PRINTER"
           SET DEVICE TO SCREEN
           ? "PRINTER OK.
        ELSE
           ? "PRINTER IS NOT READY"
        ENDIF

&  3.112 ISUPPER()
! รูปแบบ
ISUPPER(<ขอความ>) --> ผลการตรวจสอบ
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ตรวจสอบวาอักษรตัวแรกเปนตัวใหญหรือไม

: ตัวอยางที่ 3.134
        ? ISALPHA("aBCD")       // .F.  IT WILL CHECK THE FIRST COL.
        ? ISALPHA("aaaa")       // .F.
        ? ISALPHA("1ABB")       // .F.
        ? ISALPHA("1234")       // .F.
        ? ISALPHA("12.4")       // .F.
        ? ISALPHA("A.a.")       // .T.

&  3.113 I2BIN()
! รูปแบบ
I2BIN(<จ ํานวนเลข>) --> เลขในรูปแบบ 16 บิต
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ แปลงตัวเลขเปนเลขในแบบ 16 บิต ซึ่งใชเนื้อที่ 2 ไบต (8*2=16)
: ตัวอยางที่ 3.135
        ? ASC(SUBSTR(I2BIN(1),1,1))     // 1
        ? ASC(SUBSTR(I2BIN(1),2,1))     // 0
        ? ASC(SUBSTR(I2BIN(2),1,1))     // 2
        ? ASC(SUBSTR(I2BIN(2),2,1))     // 0
        ? ASC(SUBSTR(I2BIN(20001),1,1)) // 33
        ? ASC(SUBSTR(I2BIN(20001),2,1)) // 78   โดย 20001=78*256+33
        ? ASC(SUBSTR(I2BIN(20002),1,1)) // 34
        ? ASC(SUBSTR(I2BIN(20002),2,1)) // 78   โดย 20002=78*256+34

&  3.114 LASTKEY()
! รูปแบบ
LASTKEY() --> คาของแปนพิมพที่ถูกกด
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ตรวจสอบการกดแปนพิมพครั้งสุดทาย
: ตัวอยางที่ 3.136
        SALARY = 0
        @ 5,10 SAY "GET SALARY : " GET SALARY
        READ
        IF LASTKEY() = 27

           ? "NO SALARY"
        ELSE
           TAX = SALARY * 0.07
           ? TAX
        ENDIF

&  3.115 LASTREC()
! รูปแบบ
LASTREC() | RECCOUNT() --> จ ํานวนเรคอรดของแฟม
P หนาที่
ฟงกชนันี้จะท ําหนาที่ บอกจํ านวนของเรคอรดในแฟม ใหผลเหมือน RECCOUNT()
: ตัวอยางที่ 3.137
        USE FILEB
        USE FILEA
        ? LASTREC()             //        4
        ? RECCOUNT()            //        4
        SET FILTER TO FIELD2 >= 800
        ? RECCOUNT()            //        4
        ? LASTREC()             //        4
        COUNT TO CNT
        ? CNT                   //        3
        ? FILEB->(LASTREC())    //       20

&  3.116 LEFT()
! รูปแบบ
LEFT(<ขอความ>, <จํ านวนอักขระ>) --> อักขระที่ถูกแยกออกมา
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สงคาทางซายของตัวอักษรเทาที่ระบุ
: ตัวอยางที่ 3.138
        A = "YONOK COLLEGE"
        ? LEFT(A,5)             // YONOK
        B = "SUPERMAN"
        ? B +" OF "+ LEFT(A,5)  // SUPERMAN OF YONOK

&  3.117 LEN()
! รูปแบบ
LEN(<ขอความ>) --> ความยาว
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สงความยาวของตัวอักษรที่ตองการ
: ตัวอยางที่ 3.139
        A = "YONOK COLLEGE"
        ? LEN(A)                //        13
        B = LEN("SUPERMAN")     //         8
        ? B +" OF "+ LEFT(A,B)  //         8 OF YONOK CO

&  3.118 LOG()
! รูปแบบ
LOG(<จ ํานวนเลข>) --> ผลการค ํานวณ
P หนาที่
ฟงกชนันี้จะท ําหนาที่ คํ านวณคา NATURAL LOGARITHM
: ตัวอยางที่ 3.140
        ? LOG(10)               //         2.3
        ? LOG(10 * 2)           //         3.0
        ? LOG(2.71)             //         1.0
        ? LOG(EXP(10))          //        10.0

&  3.119 LOWER()
! รูปแบบ
LOWER(<ขอความ>) --> ตัวอักษรที่เปนตัวเล็กหมด
P หนาที่
ฟงกชนันี้จะท ําหนาที่ แปลงอักษรตัวใด ๆ เปนตัวใหญ
: ตัวอยางที่ 3.141
        X = "superMAN"
        ? LOWER(X)              // superman
        ? LOWER("aABBCc")       // aabbcc
        ? LOWER("a1a2Aa")       // a1a2aa
        ? LEN(LOWER("123ABAB")  // 7

&  3.120 LTRIM()
! รูปแบบ
LTRIM(<ขอความ>) --> อักขระที่ถูกตัดชองวางทางซาย
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ เอาชองวางดานซายของตัวอักษรออกไป
: ตัวอยางที่ 3.142
        X = "  ABCDEF"
        Y = "ABC"
        ? X                     // ABCDEF
        ? Y                     // ABC
        Z = 12
        ? Z                     //         12
        ? LEN(STR(Z))           //         10
        ? LTRIM(STR(Z+2))       // 14
        ? LTRIM(X)              // ABCDEF

&  3.121 LUPDATE()
! รูปแบบ
LUPDATE() --> วนัที่ลาสุดของแฟม
P หนาที่
ฟงกชันนี้จะท ําหนาที่ บอกวันที่ลาสุดที่มีการปรับปรุงแฟม
: ตัวอยางที่ 3.143
        ? DATE()                // 12/31/02     วันที่วันนี้
        USE FILEA
        ? LUPDATE()             // 12/15/02     วันที่ของแฟม
        DELETE
        PACK
        CLOSE ALL
        USE FILEA
        ? LUPDATE()             // 12/31/02     วันที่ของแฟมเปลี่ยนไป

&  3.122 L2BIN()
! รูปแบบ
L2BIN(<จ ํานวนเลข>) --> เลขในรูปแบบ 32 บิต

P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ แปลงตัวเลขเปนเลขในแบบ 32 BIT ซึ่งใชเนื้อที่ 4 ไบต (8*4=32)
: ตัวอยางที่ 3.144
        ? ASC(SUBSTR(L2BIN(1),1,1))     // 1
        ? ASC(SUBSTR(L2BIN(1),2,1))     // 0
        ? ASC(SUBSTR(L2BIN(1),3,1))     // 0
        ? ASC(SUBSTR(L2BIN(1),4,1))     // 0
        ? ASC(SUBSTR(L2BIN(2),1,1))     // 2
        ? ASC(SUBSTR(L2BIN(2),2,1))     // 0
        ? ASC(SUBSTR(L2BIN(2),3,1))     // 0
        ? ASC(SUBSTR(L2BIN(2),4,1))     // 0
        ? ASC(SUBSTR(L2BIN(10001),1,1)) // 17   อีก 3 ไบตคือ 39 0 0
        ? ASC(SUBSTR(L2BIN(10001),2,1)) // 39
        ? ASC(SUBSTR(L2BIN(10002),1,1)) // 18   อีก 3 ไบตคือ 39 0 0
        ? ASC(SUBSTR(L2BIN(10002),2,1)) // 39   โดย 10002=39*256+18

&  3.123 MAX()
! รูปแบบ
MAX(<นพิจนที ่1>, <นิพจนที ่2>) --> นิพจนที่มีคามากที่สุด
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สงคาที่มากที่สุดของการเปรียบเทียบเลข 2 จํ านวน
: ตัวอยางที่ 3.145
        A = 12
        B = 15 ;     C = 0
        ? MAX(A,B)              //         15
        ? DATE()                // 12/31/96
        ? MAX(DATE(),DATE()+2)  // 01/02/97
        ? MAX(3,12)             //         12

&  3.124 MAXCOL()
! รูปแบบ
MAXCOL() --> ตวัเลข
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สงคาหลักมากสุดที่เปนไปไดของจอ

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


136                                           การเขียนโปรแกรมส ําหรับฐานขอมูลดวย Clipper                                                         

: ตัวอยางที่ 3.146
        X = MAXCOL()
        ? X                     // 80
        @ 0,0 TO 10,MAXCOL()    // TO WRITE BOX

&  3.125 MAXROW()
! รูปแบบ
MAXROW() --> ตวัเลข
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สงคาแถวมากที่สุดที่เปนไปไดของจอ
: ตัวอยางที่ 3.147
        X = MAXROW()
        Y = MAXCOL()
        ? X,Y                    //     25      80
        DBEDIT(1,1,X+1,Y-1)

&  3.126 MEMOEDIT()
! รูปแบบ
MEMOEDIT([<ขอความที่ตองการแกไข>],
         [<มมุบน>], [<มุมซาย>], [<มุมลาง>], [<มุมขวา>])
         --> ขอความที่ผานการแกไข
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แสดงหรือแกไขฟลดแบบ MEMO
ปุมที่ใชใน MEMOEDIT()
UPARROW         เลื่อนเคอรเซอรขึ้น CTRL-PGUP       ไปที่หนาแรกของบันทึก หรือแฟมขอความนี้
DNARROW         เลื่อนเคอรเซอรลง CTRL-PGDN       ไปที่หนาสุดทายของบันทึก หรือแฟมขอความนี้
LEFTARROW      เลื่อนเคอรเซอรไปทางซาย RETURN          ไปตนบรรทัดของบรรทัดตอไป
RIGHTARROW    เลื่อนเคอรเซอรไปทางขวา DELETE          ลบตัวอักษรที่ต ําแหนงเคอรเซอร
CTRL-LEFT       เลื่อนเคอรเซอรไปทางซาย 1 คํ า BACKSPACE       ลบตัวอักษรหนาตํ าแหนงเคอรเซอร
CTRL-RIGHT      เลื่อนเคอรเซอรไปทางขวา 1 คํ า CTRL-Y          ลบบรรทัดปจจุบันที่เคอรเซอรอยู
HOME            ไปที่ตนบรรทัด CTRL-T          ลบขอความที่ตอทายเคอรเซอรไปจนหมดบรรทัด
END             ไปที่ทายบรรทัด CTRL-B          จดัรูปแบบของยอหนาใหม
CTRL-HOME       ไปที่บรรทัดแรกของจอภาพ CTRL-V/INS      สั่งใหปุม INSERT อยูในสถานะเปนจริง
CTRL-END        ไปที่บรรทัดสุดทายของจอภาพ CTRL-W          ยอมรับการแกไข แลวออกจากฟงกชันนี้
PGUP            ไปที่หนากอนหนานี้ ESC             ยกเลิกการแกไข แลวออกจากฟงกชันนี้
PGDN            ไปที่หนาตอไป

: ตัวอยางที่ 3.148
        TXT = "ABCDE"+CHR(13)+CHR(10)+"FGHIJ"+CHR(13)+CHR(10)+ ;
              "KLMNO"+CHR(13)+CHR(10)+"PQRST"
        SET CURSOR OFF
        MEMOEDIT(TXT,10,15,20,60,.F.)  // อานขอความมาดู แตไมแกไข
        SET CURSOR ON
: ตัวอยางที่ 3.149
        CLS
        TXT = "ABCDE"+CHR(13)+CHR(10)+"FGHIJ"+CHR(13)+CHR(10)+ ;
              "KLMNO"+CHR(13)+CHR(10)+"PQRST"
        SETCOLOR("W+/B")
        TXT = MEMOEDIT(TXT,10,15,20,60)  // อานขอความมาปรับปรุง
        CLS
        FOR I = 1 TO MLCOUNT(TXT)
            ?? MEMOLINE(TXT,80,I)
        NEXT
: ตัวอยางที่ 3.150
        CLS
        TXT = MEMOREAD("C:\AUTOEXEC.BAT")
        SETCOLOR("W+/B")
        TXT = MEMOEDIT(TXT,10,15,20,60)  // อานแฟมขอความมาปรับปรุง
        CLS
        FOR I = 1 TO MLCOUNT(TXT)        // ปรับปรุงลง TXT
            ?? MEMOLINE(TXT,80,I)        // ไมเขียนลง C:\AUTOEXEC.BAT
        NEXT
: ตัวอยางที่ 3.151
        CLS
        TXT = MEMOREAD("C:\TEST.TXT")
        SETCOLOR("W+/B")
        TXT = MEMOEDIT(TXT,10,15,20,60) // อานแฟมขอความมาปรับปรุง
        MEMOWRIT("C:\TEST.TXT",TXT)   // ถาไมเคยมีแฟมนี้มากอน
        CLS                              // ค ําสั่งนี้จะสรางแฟมใหมให
        FOR I = 1 TO MLCOUNT(TXT)      // ปรับปรุงลง TXT และ
            ?? MEMOLINE(TXT,80,I)        // เขียนลง C:\TEST.TXT

        NEXT
        // แฟมที่สรางดวย MEMOEDIT จะมีรหัส 1A ปดทายแฟมเสมอ

&  3.127 MEMOLINE()
! รูปแบบ
MEMOLINE(<ขอความ>,
         [<ขนาดของบรรทัดที่ตองการแสดง>],
         [<บรรทดัทีต่องการแสดง>]) --> ขอความที่ถูกเลือก
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ น ําคาในฟลดแบบ MEMO มาใชโดยเลือกบรรทัดที่ตองการ
: ตัวอยางที่ 3.152
        TXT = "ABCDE"+CHR(13)+CHR(10)+"FGHIJ"+CHR(13)+CHR(10)+ ;
              "KLMNO"+CHR(13)+CHR(10)+"PQRST"
        ?? MEMOLINE(TXT,80,1)           // ABCDE
        ?? MEMOLINE(TXT,80,2)           // FGHIJ
        ?? MEMOLINE(TXT,80,3)           // KLMNO
        ?? MEMOLINE(TXT,80,4)           // PQRST
        INKEY(0)  // แตละบรรทัดมีรหัสตัดบรรทัดของตนเอง จึงไมตองใช ?

&  3.128 MEMOREAD()
! รูปแบบ
MEMOREAD(<ชือ่แฟม>) --> ขอความที่รับคาจากแฟม
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ การอานคาแฟมขอความทั้งแฟม มาเก็บในตัวแปรแบบ MEMO
: ตัวอยางที่ 3.153
        TXT = MEMOREAD("TEST.TXT")
        FOR I = 1 TO MLCOUNT(TXT)
            ? MEMOLINE(TXT,80,I)    // อานขอความจากแฟมมาแสดงผล
        NEXT

&  3.129 MEMORY()
! รูปแบบ
MEMORY(<ระบพุืน้ที>่) --> จํ านวนกิโลไบตที่เหลือใช
P หนาที่

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


                                                                    บทที่ 3  :  ฟงกชัน (Function)                                                              139

ฟงกชนันี้จะท ําหนาที่ แสดงหนวยความจํ าที่เปนไปได
: ตัวอยางที่ 3.154
        ? MEMORY(2)             // บอกหนวยความจํ าสูงสุดที่ใชคํ าสั่ง RUN ได
        ? MEMORY(1)             // พื้นที่เปน KB ที่เหลือส ําหรับโปรแกรม (64KB)
        ? MEMORY(0)             // พื้นที่สูงสุด ที่เหลือทั้งหมด
        IF MEMORY(2) >= 128
           RUN C:\WORD\CW\CW.EXE
        ELSE
           ? "NOT ENOUGH MEMORY TO RUN PROGRAM"
        ENDIF

&  3.130 MEMOTRAN()
! รูปแบบ
MEMOTRAN(<ขอความ>,
         [<ตวัอักษรที่ใชแทนที่ HARD CR>],
         [<ตวัอกัษรทีใ่ชแทนที ่SOFT CR>]) --> ขอความถูกปรับปรุง
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ แทนคา CARRIAGE RETURN/LINE FEEDS ดวยตัวอักษรตามตองการ
: ตัวอยางที่ 3.155
        TXT = MEMOREAD("TEST.TXT")    ;        TXT = MEMOTRAN(TXT,"!","&")
        FOR I = 1 TO MLCOUNT(TXT)
            ? MEMOLINE(TXT,80,I)    // แตแฟม TEST.TXT จะไมถูกปรับปรุง
        NEXT
        // HARD CARRIAGE RETURN คอื ผลของการกด ENTER ในโปรแกรมทั่วไป
        // SOFT CARRIAGE RETURN คอื ผลของโปรแกรมที่ปดบรรทัดใหกับแฟมขอความ
        // รหัส ASCII ของ HARD CARIAGE RETURN คือ 13
        // รหัส ASCII ของ SOFT CARIAGE RETURN คือ 141

&  3.131 MEMOWRIT()
! รูปแบบ
MEMOWRIT(<ชือ่แฟม>, <ขอความที่เขียนลงแฟม>) --> ผลการเขียนลงแฟม
P หนาที่
ฟงกชันนี้จะท ําหนาที่ เขียนแฟมขอความ หรือ MEMO FIELD ลงไปในแฟม

: ตัวอยางที่ 3.156
        TXT = MEMOREAD("C:\AUTOEXEC.BAT")
        MEMOWRIT("TEST.TXT",TXT)       // น ําขอความในแฟม เขียนลงแฟม
        CNTLINE = MLCOUNT(MEMOREAD("TEST.TXT"))   ;    ?
        FOR I = 1 TO CNTLINE
            ?? MEMOLINE(MEMOREAD("TEST.TXT"),80,I)
        NEXT
: ตัวอยางที่ 3.157
        USE FILEHIST
        DO WHILE .NOT. EOF()
           IF MEMOWRIT("MEM.TXT",NOTES)
              ? "VALID TO FILE"
              RUN TYPE MEM.TXT
              INKEY(0)
           ELSE
              ? "INVALID TO FILE"
           ENDIF
           SKIP
        ENDDO

&  3.132 MIN()
! รูปแบบ
MIN(<นพิจนที ่1>, <นิพจนที ่2>) --> นิพจนที่มีคานอยที่สุด
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สงคาที่นอยที่สุดของการเปรียบเทียบเลข 2 จํ านวน
: ตัวอยางที่ 3.158
        A = 12
        B = 15
        C = 0
        ? MIN(A,B)              //         12
        ? DATE()                // 12/31/96
        ? MIN(DATE(),DATE()+2)  // 12/31/96
        ? MIN(3,12)             //          3

&  3.133 MLCOUNT()
! รูปแบบ
MLCOUNT(<ขอความ>) --> จํ านวนบรรทัดของแฟมขอความ
P หนาที่
ฟงกชนันี้จะท ําหนาที่ นับจ ํานวนบรรทัดในแฟมขอความ หรือ MEMO FIELD
: ตัวอยางที่ 3.159
        TXT = MEMOREAD("C:\AUTOEXEC.BAT")
        CNTLINE = MLCOUNT(TXT)
        ? "THIS FILE HAVE "+ STR(CNTLINE)
        ? "---------------------"
        ?                                 // ตองตัดบรรทัดให
        FOR I = 1 TO CNTLINE              // จงึจะพิมพที่หลักแรกไดถูกตอง
            ?? MEMOLINE(TXT,80,I)
        NEXT

&  3.134 MLCTOPOS()
! รูปแบบ
MLCTOPOS(<ขอความ>, <ความกวางของบรรทัด>, <จํ านวนบรรทัด>, <ตํ าแหนงหลัก>)
         --> ตํ าแหนงลาสุด
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สงตํ าแหนงของขอมูลเปนตํ าแหนงที่ โดยบอกเปนแถวและหลัก
โดยการนบัแถวจะเริ่มจาก 1 แตการนับหลักจะเริ่มจาก 0 ของแตละแถว
ตรงขามกับฟงกชัน MPOSTOLC()
: ตัวอยางที่ 3.160
        TXT = 'ASFASFASFDASFASFASFASFASDFASFASASDFSAF'
        ? MLCTOPOS(TXT,5,2,0)           // 6
        ? MLCTOPOS(TXT,6,3,0)           // 13
        ? MLCTOPOS(TXT,7,4,2)           // 24

&  3.135 MLPOS()
! รูปแบบ
MLPOS(<ขอความ>, <ความกวางของบรรทัด>, <จํ านวนบรรทัด>) --> ตํ าแหนงลาสุด

P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ระบุต ําแหนงแรกของบรรทัดที่ตองการในแฟมขอความ หรือ MEMO FIELD
: ตัวอยางที่ 3.161
        TXT = MEMOREAD("C:\AUTOEXEC.BAT")
        CHRINLINE = 40      // ยิง่มากยิ่งด ีแตถานอยจะมีผลตอการนับบรรทัด
        CNTLINE = 3         // ตองการทราบวาต ําแหนงแรกของบรรทัดที ่3
        X = MLPOS(TXT,CHRINLINE,CNTLINE)
        ? X       // 17 หมายถึงตัวเลขแสดงวาบรรทัดที ่3 เริ่มไบตที ่17
        ? SUBSTR(TXT,X,20)  // น ําอกัษร 20 ตัวจากบรรทัดที่ 3 มาแสดง

&  3.136 MOD()
! รูปแบบ
MOD(<ตวัตัง้>, <ตัวหาร>) --> เศษจากการหาร
P หนาที่
ฟงกชนันี้จะท ําหนาที่ สงเศษของการหาร ของตัวแปร 2 จํ านวน
: ตัวอยางที่ 3.162
        ? MOD(3,1)                      //      0.00
        ? MOD(5,2)                      //      1.00
        ? MOD(101,20)                   //      1.00

&  3.137 MONTH()
! รูปแบบ
MONTH(<วนัที่>) --> เลขเดือน
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แปลจาก DATE ใหเปนเลขเดือน
: ตัวอยางที่ 3.163
        ? DATE()                        // 12/31/96
        ? MONTH(DATE())                 // 12
        ? MONTH(CTOD("05/113/95"))      // 05

&  3.138 MPOSTOLC()
! รูปแบบ
MPOSTOLC(<ขอความ>, <ความกวาง>, <ตํ าแหนงหลัก>) --> ชุดอาเรยเก็บบรรทัดและหลัก

P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สงคืนคาตํ าแหนงของขอมูลเปนแถวและหลัก เก็บในอาเรย โดยบอกตํ าแหนงที่
ตรงขามกับฟงกชัน MLCTOPOS()
: ตัวอยางที่ 3.164
        TXT ='OAKALKLJKHLKJHLKJDFJAAFJKFJASFASF'
        X = MPOSTOLC(TXT,5,6)
        ? X[1],X[2]                                              //   2  0
        ? MPOSTOLC(TXT,5,0)[1],MPOSTOLC(TXT,5,0)[2]     //   0  0
        ? MPOSTOLC(TXT,5,1)[1],MPOSTOLC(TXT,5,1)[2]     //   1  0
        ? MPOSTOLC(TXT,5,2)[1],MPOSTOLC(TXT,5,2)[2]     //   1  1
        ? MPOSTOLC(TXT,5,3)[1],MPOSTOLC(TXT,5,3)[2]     //   1  2
        ? MPOSTOLC(TXT,5,4)[1],MPOSTOLC(TXT,5,4)[2]     //   1  3
        ? MPOSTOLC(TXT,5,5)[1],MPOSTOLC(TXT,5,5)[2]     //   1  4
        ? MPOSTOLC(TXT,5,7)[1],MPOSTOLC(TXT,5,7)[2]     //   2  1
        ? MPOSTOLC(TXT,5,8)[1],MPOSTOLC(TXT,5,8)[2]     //   2  2
        ? MPOSTOLC(TXT,5,9)[1],MPOSTOLC(TXT,5,9)[2]     //   2  3
        // ถาระบคุวามกวางเปน 5 แลวตองการหาวาตํ าแหนงที่ 9 อยูที่ใด
        // จะไดค ําตอบเปนแถวที่ 2 หลักที ่3 เพราะการจัดเรียงขอมูลเปนตาราง
        // ต ําแหนงแรกอยูแถวที ่1 หลักที ่0 หรือต ําแหนงที่หกอยูแถวที่ 2 หลักที ่0

&  3.139 NEXTKEY()
! รูปแบบ
NEXTKEY() --> คาของแปนพิมพที่ถูกกด
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ตรวจสอบการกดแปนพิมพเหมือน INKEY() แต INKEY() จะลบคาในหนวยความจํ า
NEXTKEY() จะอานคามาแตไมลบคาออกจากหนวยความจํ า
ระหวาง INKEY() LASTKEY() และ NEXTKEY() มีการทํ างานรวมกันไดดีมาก
INKEY()   เมือ่ไดคาจากหนวยความจ ําวากดอะไร จะสงคาไปยัง LASTKEY()
LASTKEY() จะตรวจสอบวาคาสุดทายในหนวยความจํ า มักรับคามาจาก INKEY()
NEXTKEY() อานคาจากหนวยความจ ํา แตจะไมเปลี่ยนแปลงคาในหนวยความจํ า
: ตัวอยางที่ 3.165
        CLS
        KEYBOARD CHR(65)
        ? NEXTKEY(),NEXTKEY() //      65      65

        ? LASTKEY(),INKEY()     //       0      65
        ? INKEY(),LASTKEY()     //       0      65
        ? LASTKEY(),NEXTKEY()   //      65       0
        ACCEPT TO X             // ABCD แลวกด ENTER
        ? X                     // ABCD
        ? NEXTKEY(),INKEY()     //       0       0
        ? LASTKEY(),NEXTKEY() //      13       0
        WAIT                    // PRESS ANY KEY TO CONTINUE
        WAIT "=> Y/N"           // Y/N   ถากด A จะได LASTKEY() = 97
        ? NEXTKEY(),INKEY()     //       0       0
        ? LASTKEY(),NEXTKEY() //      97       0
        INKEY(0)                // รอรับคาหากกด A จะได LASTKEY() = 65
        ? NEXTKEY(),INKEY()     //       0       0
        ? LASTKEY(),NEXTKEY()  //      65       0
        @ 15,10 GET X           // ถารบัคาเปน D จะทํ าให LASTKEY() = 100
        READ
        ? NEXTKEY(),INKEY()     //       0       0
        ? LASTKEY(),NEXTKEY() //     100       0
: ตัวอยางที่ 3.166
        CLS                     // ตวัอยางที่นี้แสดงใหเปนการใช KEYBOARD
        KEYBOARD "ABCDEF"+CHR(13)+CHR(13)+CHR(65)+CHR(66)
        ? NEXTKEY(),INKEY()     //      97      97
        ? NEXTKEY(),INKEY()     //      98      98
        ? NEXTKEY(),INKEY()     //      99      99
        ? NEXTKEY(),NEXTKEY() //     100     100
        WAIT ""                 // D
        ACCEPT TO TESTKEY1   // EF
        INKEY(0)                // A
        WAIT ""                 //
        ? NEXTKEY(),LASTKEY()  //      66      65

&  3.140 OS()
! รูปแบบ
OS() --> ชือ่ระบบปฏิบัติการ
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ บอกชื่อระบบปฏิบัติการที่ก ําลังใชอยูในระบบคอมพิวเตอรของเรา

: ตัวอยางที่ 3.167
        CLS
        ? OS()                  // DOS 6.20
        FOR I = 1 TO 10 ;    ? OS()  ;        NEXT

&  3.141 OUTERR()
! รูปแบบ
OUTERR(<รายการนิพจน>) --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ แสดงขอความออกทางอุปกรณที่เตรียมไวแสดงขอผิดพลาด
: ตัวอยางที่ 3.168
        USE FILEA               // ถาเปดแฟมครั้งแรกจะไดคา RLOCK() = .T.
        IF RLOCK() = .F.        // รายงานทางอุปกรณที่แสดงขอผิดพลาด
           OUTERR("INVALID SHARED",DATE(),TIME())
        ENDIF
        USE FILEA NEW
        IF RLOCK() = .F.        // รายงานวาแฟมถูกจองไว ไมสามารถนํ ามาใชได
           OUTERR("INVALID SHARED",DATE(),TIME()) // แบบนี้แสดงทางจอภาพ
        ENDIF

&  3.142 OUTSTD()
! รูปแบบ
OUTSTD(<รายการนิพจน>) --> NIL
P หนาที่
ฟงกชนันี้จะท ําหนาที่ แสดงขอมูลออกทางอุปกรณมาตรฐาน
ถาโปรแกรมนี้ชื่อ X.EXE เมื่อประมวลผลที่ดอสวา C:\X >AA
โดยแฟม AA คือผลลัพธที่ไดจากการสั่ง OUTSTD ในโปรแกรมนี ้แต ? จะไมมีผลตอแฟม AA
: ตัวอยางที่ 3.169
        USE FILEA
        DO WHILE .NOT. EOF()
            OUTSTD(FIELD1,FIELD2,FIELD3) // ไมเลื่อนบรรทัดใหเหมือน QOUT
            ?                            // ท ําหนาที่เลื่อนบรรทัด เทานั้น
            SKIP
        ENDDO

&  3.143 PAD()
! รูปแบบ
PADL(<นพิจนทีต่องการแสดง>, <ความยาว>) --> ขอความที่ชิดซาย
PADC(<นพิจนทีต่องการแสดง>, <ความยาว>) --> ขอความที่อยูกึ่งกลาง
PADR(<นพิจนทีต่องการแสดง>, <ความยาว>) --> ขอความที่ชิดขวา
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ จัดชุดตัวอักษรใหเปนระเบียบ เชน ชิดซาย ชิดขวา หรือกึ่งกลาง
: ตัวอยางที่ 3.170
        CLS
        SETCOLOR("W+/B")
        X = PADC("DATE : "+DTOC(DATE()), 80)
        @ 21, 0 SAY X
        @ 22, 0 SAY PADR("TIME OF DAY : " + TIME(),80)
        @ 23, 0 SAY PADL("SECOND : "+ STR(SECOND()),80)
        INKEY(0)

&  3.144 PCOL()
! รูปแบบ
PCOL() --> จ ํานวนสูงสุดของหลัก
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สงคาหลักปจจุบันที่อยูบนเครื่องพิมพ คลายกับ COL() แตใชกับเครื่องพิมพ
: ตัวอยางที่ 3.171
        SET DEVICE TO PRINTER
        FOR I = 1 TO 5                      // ! *
           @ PROW()+1,1 SAY "!"             // ! * *
           FOR J = 1 TO I                   // ! * * *
               @ PROW() , PCOL()+1 SAY "*"  // ! * * * *
           NEXT                             // ! * * * * *
        NEXT                                // * จะหางกันเพราะ PCOL()+1
        EJECT                               // สัง่เลือ่นกระดาษออกจากเครื่องพิมพ
        SET DEVICE TO SCREEN

&  3.145 PCOUNT()
! รูปแบบ
PCOUNT() --> เลขแสดงการตรวจสอบการรับคาของโปรแกรมยอย
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ตรวจสอบการรับตัวแปรของโปรแกรมยอย ถาไดคา 0 หมายถึงไมมีการรับตัวแปรเขามา
: ตัวอยางที่ 3.172
        DO XX WITH (5)                  // 1
        DO XX                           // 0
        DO XX WITH ("")                 // 1
        DO XXX WITH (5,5,5)             // 1
        PROCEDURE XX (NUM)
           ? PCOUNT()
        RETURN
        PROCEDURE XXX (NUM1,NUM2,NUM3)
           ? PCOUNT()
        RETURN

&  3.146 PROCLINE()
! รูปแบบ
PROCLINE([<ล ําดบัของโปรแกรมที่พิจารณา>]) --> เลขที่บรรทัดในโปรแกรม
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ บอกเลขบรรทัดของโปรแกรมที่ใชงานอยูจนถึงคํ าสั่งปจจุบัน
: ตัวอยางที่ 3.173
        DO P1
        PROCEDURE P1
           DO P2
        RETURN
        PROCEDURE P2 ;   DO P3  ;     RETURN
        PROCEDURE P3
           ? PROCNAME(0),PROCLINE(0)    // P3     9
           ? PROCNAME(1),PROCLINE(1)    // P2     6
           ? PROCNAME(2),PROCLINE(2)   // P1     3
           ? PROCNAME(3),PROCLINE(3)    // X      1
        RETURN

: ตัวอยางที่ 3.174
        DO P1
        PROCEDURE P1
           DO P2
        RETURN
        PROCEDURE P2
           DO P3
        RETURN
        PROCEDURE P3     // ตวัอยางที่นี้ท ําใหรูวาเรียกโปรแกรมอะไรมาบาง
           I = 0
           DO WHILE LEN(PROCNAME(I)) > 0 // P3   11
              ? PROCNAME(I),PROCLINE(I)         // P2    6
              I++                               // P1    3
           ENDDO                                // X    1
        RETURN

&  3.147 PROCNAME()
! รูปแบบ
PROCNAME([<ล ําดบัของโปรแกรมที่พิจารณา>]) --> ชื่อโปรแกรมยอย
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ บอกชื่อโปรแกรมยอยที่มีการเรียกใชงานอยูเปนชั้น ๆ จนถึงชั้นนอกสุด
: ตัวอยางที่ 3.175
        DO P1                           // PROC1
        DO P2                           // PROC2
        DO P3                           // PROC3
        ? PROCNAME(0)                   // X  (CURRENT PROCEDURE)
        PROCEDURE P1
           ? "PROC1"
        RETURN
        PROCEDURE P2
           ? "PROC2"
        RETURN
        PROCEDURE P3
           ? "PROC3"

           ? PROCNAME(0)                // P3
           ? PROCNAME(1)                // X
        RETURN
: ตัวอยางที่ 3.176
        DO P1
        PROCEDURE P1
           DO P2
        RETURN
        PROCEDURE P2  ;    DO P3    ;      RETURN
        PROCEDURE P3
           ? PROCNAME(0)                // P3
           ? PROCNAME(1)                // P2
           ? PROCNAME(2)                // P1
           ? PROCNAME(3)                // X
        RETURN

&  3.148 PROW()
! รูปแบบ
PROW() --> ต ําแหนงแถวบนเครื่องพิมพ
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สงคาแถวปจจุบันที่ระบุบนเครื่องพิมพ คลายกับ ROW() แตใชกับเครื่องพิมพ
: ตัวอยางที่ 3.177
        USE FILEA
        SET DEVICE TO PRINTER
        DO WHILE .NOT. EOF()
           @ PROW()+1,10 SAY FIELD1
           @ PROW()  ,20 SAY FIELD2
           @ PROW()  ,30 SAY FIELD3
           SKIP
        ENDDO
        EJECT
        SET DEVICE TO SCREEN

&  3.149 QOUT()
! รูปแบบ
QOUT([<รายการนิพจน>]) --> NIL
QQOUT([<รายการนิพจน>]) --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ แสดงขอมูลออกทางจอภาพ(CONSOLE) โดยมีการตัดบรรทัดใหอัตโนมัติทุกครั้ง
QOUT  พมิพแตละเรคอรดจะตัดบรรทัดให คลายคํ าสั่ง ?
QQOUT พมิพแตละเรคอรดจะไมตัดบรรทัดให คลายคํ าสั่ง ??
: ตัวอยางที่ 3.178
        X = {1,2}
        XXX = {|Y|QOUT(Y)}
        AEVAL (X,{|Y|QOUT(Y)})                  //  1
        X = {"A = A +1",120,.T.,"THAILAND"}   //  2
        AEVAL (X,XXX)             // การน ําขอมูลในอาเรยมาแสดง
        Z = ARRAY(4)
        Z[1] = [RED]
        Z[2] = [GREEN]
        Z[3] = [YELLOW]
        AEVAL (Z,XXX)            // อาเรยตัวที่ 4 ไมมีจะออกคํ าวา NIL
: ตัวอยางที่ 3.179
        USE FILEA
        DO WHILE .NOT. EOF()
            QOUT(FIELD1,FIELD2,FIELD3) // เลื่อนบรรทัดใหอัตโนมัติ
            SKIP
        ENDDO

&  3.150 RAT()
! รูปแบบ
RAT(<ตวัอกัษรทีต่องการน ําไปคน>, <ขอความเปาหมาย>) --> ตํ าแหนงที่คนพบ
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ สงตํ าแหนงสุดทายที่พบตัวอักษรในชุดตัวอักษรที่ตองการหา
: ตัวอยางที่ 3.180
        X = "ABCDEFABCDEF"

        ? RAT("E",X)                    // 11
        X = "C:\WINDOWS\WIN.COM"
        POS = RAT("\",X)
        ? POS                           // 11
        ? LEN(X)                        // 18
        ? SUBSTR(X,POS+1,LEN(X)-POS)    // WIN.COM
        ? SUBSTR(X,1,POS)               // C:\WINDOWS\

&  3.151 READEXIT()
! รูปแบบ
READEXIT([<ตรรกนพิจน>]) --> ตรรกนิพจนแสดงคาปจจุบัน
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ตรวจสอบการออกจากคํ าสั่ง READ ดวยลูกศรขึ้นลง หรือการสั่งอนุญาติใหออกหรือไมออก
: ตัวอยางที่ 3.181
        CLS
        X = SPACE(5)
        Y = SPACE(5)
        Z = SPACE(5)
        ? READEXIT()            // .F. ไมใหออกจาก READ ดวยลูกศร
        OLDMODE = READEXIT()    // เก็บคาเดิมไว
        READEXIT(.T.)           // คอืการอนุญาติใหออกจาก READ ดวยลูกศร
        @ 4,10 SAY "TEST ARROW ON READ " GET X
        @ 5,10 SAY "TEST ARROW ON READ " GET Y
        @ 6,10 SAY "TEST ARROW ON READ " GET Z
        READ                    // อนญุาติใหเลิกดวยลูกศรได
        READEXIT(OLDMODE)       // เซตคาใหมตามคาที่เคยเก็บไว
        @ 7,10 SAY "TEST ARROW ON READ " GET X
        @ 8,10 SAY "TEST ARROW ON READ " GET Y
        @ 9,10 SAY "TEST ARROW ON READ " GET Z
        READ                    // ไมอนุญาติใหเลิก READ ดวยลูกศร

&  3.152 READINSERT()
! รูปแบบ
READINSERT([<ตรรกนพิจน>]) --> ตรรกนิพจนแสดงคาปจจุบัน

P หนาที่
ฟงกชันนี้จะท ําหนาที่ ตรวจสอบการกดปุม INSERT หรือ สั่งเปดปุม INSERT
: ตัวอยางที่ 3.182
        CLS
        X = 0
        ? READINSERT()                  // .F. เสมอ เมื่อเริ่มโปรแกรม
        OLDMODE = READINSERT()
        READINSERT(.T.)
        @ 5,10 SAY "TEST INSERT ON READ " GET X
        READ
        ? READINSERT()                  // .T.
        READINSERT(.F.)                 // สั่งปด INSERT
        @ 6,10 SAY "TEST INSERT ON READ " GET X
        READ
        READINSERT(OLDMODE)             // คนืสถานะการกด INSERT
        @ 7,10 SAY "TEST INSERT ON READ " GET X
        READ

&  3.153 READKEY()
! รูปแบบ
READKEY() --> คาของแปนพิมพที่ถูกกด
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ตรวจสอบการออกจากคํ าสั่ง GET ... READ

ตารางคาที่สงคามาให READKEY() ซึ่งเปน EXIT CODE
UPARROW      5               ESC             12
DNARROW       24               CTRL-END,CTRL-W 14
PGUP            6               TYPE PAST END   15
PGDN            7               RETURN          15
CTRL-PGUP      31               CTRL-PGDN   30

: ตัวอยางที่ 3.183
        CLS
        X = 0
        @ 5,10 SAY "TEST READKEY() " GET X
        READ

        ? READKEY()             // บอกคาที่เกิดจากการหยุดคํ าสั่ง READ
        ? LASTKEY()             // คาของ LASTKEY คือการกดปุมครั้งสุดทาย

&  3.154 READVAR()
! รูปแบบ
READVAR() --> ชือ่ตวัแปรที่ก ําลังถูกเปดอยู
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ บอกชื่อตัวแปรปจจุบันที่กํ าลังทํ างานอยู สามารถนํ าไปสรางสวนชวยเหลือไดอยางดี
: ตัวอยางที่ 3.184
        CLS
        SET KEY 28 TO HELPFUNCTION      // F1
        _NAME = SPACE(20)
        _AGE  = 0
        _SALARY = 0
        @ 5,10 SAY "NAME   : " GET _NAME
        @ 6,10 SAY "AGE    : " GET _AGE
        @ 7,10 SAY "SALARY : " GET _SALARY
        READ
        FUNCTION HELPFUNCTION
        LOCAL SCR1 := SAVESCREEN(0,0,24,79)
        DO CASE
           CASE READVAR() = "_NAME"
                ?? "IT SHOULD HAVE MR. OR MS. AT THE FONT OF NAME"
           CASE READVAR() = "_AGE"
                ?? "IT SHOULD BE 0 - 99"
           CASE READVAR() = "_SALARY"
                ?? "IT SHOULD > 1000 AND < 1,000,000"
        ENDCASE
        INKEY(0)
        RESTSCREEN(0,0,24,79,SCR1)
        RETURN

&  3.155 RECCOUNT()
! รูปแบบ
RECCOUNT() | LASTREC() --> จ ํานวนเรคอรดของแฟม
P หนาที่
ฟงกชนันี้จะท ําหนาที่ บอกจํ านวนของเรคอรดในแฟม ใหผลเหมือน LASTREC()
: ตัวอยางที่ 3.185
        USE FILEA
        ? RECCOUNT()                    //          4
        ? LASTREC()                     //          4
        SET FILTER TO FIELD2 > 1000
        COUNT TO FILTER_COUNT
        ? FILTER_COUNT                  //          1
        ? RECCOUNT()                    //          4
        ? LASTREC()                     //          4

&  3.156 RECNO()
! รูปแบบ
RECNO() --> เลขที่เรคอรดปจจุบัน
P หนาที่
ฟงกชนันี้จะท ําหนาที่ บอกล ําดับของเรคอรดปจจุบันที่อยูในแฟม การ INDEXED แฟมไมมีผลกับล ําดับ
: ตัวอยางที่ 3.186
        USE FILEA
        ? RECNO()                       //          1
        GO BOTTOM
        ? RECNO()                       //          4

&  3.157 RECSIZE()
! รูปแบบ
RECSIZE() --> ขนาดของเรคอรด
P หนาที่
ฟงกชันนี้จะท ําหนาที่ บอกความยาวของ 1 เรคอรด

: ตัวอยางที่ 3.187
        USE FILEA
        ? RECSIZE()                     //         31
        ? LASTREC()                     //          4
        ? HEADER()                      //        130
        FSIZE = RECSIZE()*LASTREC()+HEADER()+1
        ? "FILE SIZE = " + STR(FSIZE,4) // FILE SIZE =  255

&  3.158 REPLICATE()
! รูปแบบ
REPLICATE(<ขอความ>, <จํ านวนทํ าซํ้ า>) --> ขอความที่ผานการทํ าซํ้ า
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ฟงกชันที่ใชคัดลอกตัวอักษร ใหไดหลาย ๆ ชุด
: ตัวอยางที่ 3.188
        ? REPLICATE("A",3)      // AAA
        ? REPLICATE("AB ",2)    // AB AB
        X = 5
        ? REPLICATE(STR(X,1),3) // 555
        FOR I = 1 TO 3          // *
            ? REPLICATE("*",I)  // **
        NEXT                    // ***

&  3.159 RESTSCREEN()
! รูปแบบ
RESTSCREEN(<มมุบน>, <มุมซาย>, <มุมลาง>, <มุมขวา>,
           <ตวัแปรเก็บจอภาพ>) --> NIL
P หนาที่
ฟงกชันนี้จะท ําหนาที่ คืนผลการ SAVESCREEN ใหจอภาพ
: ตัวอยางที่ 3.189
        SETCOLOR("GR+/B")               // ตวัอยางที่นี้จะใชพักจอภาพได
        CLS                             // ทกุ 0.1วินาทีจะพิมพขอความ
        SETPOS(5,10)                    // ทีไ่มซํ ้ากับตํ าแหนงเดิม
        ?? "CLIPPER FOR DOS"
        SCR1 = SAVESCREEN(5,10,5,24)

        DO WHILE LASTKEY() != 27
           X = MOD(SECOND()*1000,24)
           Y = MOD(SECOND()*1000,65)
           RESTSCREEN(X,Y,X,Y+14,SCR1)
           INKEY(0.1)
        ENDDO

&  3.160 RIGHT()
! รูปแบบ
RIGHT(<ขอความ>, <จํ านวนอักขระ>) --> อักษรที่ถูกแยกออกมา
P หนาที่
ฟงกชนันี้จะท ําหนาที่ เลือก SUBSTRING จากทางขวา
: ตัวอยางที่ 3.190
        ? RIGHT("ABCD",2)       // CD
        ? RIGHT("1234",2)       // 34
        X = "A1B2"
        Y =[A2B3]
        ? RIGHT(X,2)+RIGHT(Y,2) // B2B3
        ? RIGHT(X+Y,4)          // A2B3

&  3.161 RLOCK()
! รูปแบบ
RLOCK() --> ผลการจองเรคอรด
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ตรวจสอบการจองเรคอรดปจจุบัน เพื่อท ําการ ลบ เรียกคืน หรือเขียนทับลงไป
: ตัวอยางที่ 3.191
        USE FILEA
        USE FILEA NEW
        IF RLOCK()
           ? "LOCK OK."   ;        ELSE
           ? "CAN NOT LOCK."            // CAN NOT LOCK.
        ENDIF
        SELE 1   ;        CLOSE     // ปดแฟมในพื้นที ่1
        SELE 2

        IF RLOCK()                      // ตรวจสอบการจองเรคอรด
           ? "LOCK OK."
        ELSE
 ? "CAN NOT LOCK."            // CAN NOT LOCK.
        ENDIF
        USE FILEA                       // เปดแฟมนี้ใหมในพื้นที ่2
        IF RLOCK()                      // ตรวจสอบการจองเรคอรด
           ? "LOCK OK."                 // LOCK OK.  การจองเรคอรดถูกตอง
           DELETE                       // สามารถ DELETE , RECALL หรือ REPLACE
        ELSE
           ? "CAN NOT LOCK."
        ENDIF
: ตัวอยางที่ 3.192
        USE FILEA                       // โปรแกรมนี้จะไมน ําเรคอรดมาแสดง
        USE FILEA NEW                   // เพราะเรคอรดถูกจองไว
        ? RLOCK()                       // .F.
        DO WHILE .NOT. EOF()            // การวนลูปจึงน ําเรคอรดออกมาไมได
           ? FIELD1
           SKIP
        ENDDO
: ตัวอยางที่ 3.193
        USE FILEA
        USE FILEA NEW
        SELE 1                          // เปดแฟมแบบนี้จะน ําเรคอรด
        ? RLOCK()                       // .T.
        DO WHILE .NOT. EOF()            // มาแสดงไดเพราะการจองถูกใช
           ? FIELD1                     // ในพื้นที ่ที่จองอยางถูกตอง
           SKIP
        ENDDO
: ตัวอยางที่ 3.194
        USE FILEA SHARED                // เปดแฟมแบบนี ้ท ําใหพื้นที่อื่นใชแฟมรวมได
        USE FILEA SHARED NEW
        ? RLOCK()                       // .T.
        DO WHILE .NOT. EOF()

           ? FIELD1
           SKIP
        ENDDO

&  3.162 ROUND()
! รูปแบบ
ROUND(<ตวัเลข>, <จํ านวนทศนิยม>) --> ตัวเลขที่ผานการปดเศษ
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ปดเศษใหกับตัวแปรตัวเลข
: ตัวอยางที่ 3.195
        ? ROUND(101.1,0)       // 0 หมายถึงจ ํานวนทศนิยม จะได 101
        ? ROUND(106.1,-1)      // ปดเศษตั้งแตหลักสิบ จะได 110
        ? ROUND(106.1,-2)      // ปดเศษตั้งแตหลักรอย จะได 100
        ? ROUND(106.1,-1)      // 110
        ? ROUND(106.1,-2)      // 100
        ? ROUND(106.164,2)     // 106.16
        ? ROUND(106.164,1)     // 106.2
        ? ROUND(106.164,3)     // 106.164
        SET DECIMALS TO 3
        SET FIXED ON
        ? ROUND(106.164,1)     // 106.200
        ? ROUND(106.164,-1)    // 110.000 (-1 หมายถงึไมเอาหนาจุด 1 หลัก)
        ? ROUND(106.164,2)     // 106.160 ( 2 หมายถงึเอาหลังจุด 2 หลัก)

&  3.163 ROW()
! รูปแบบ
ROW() --> เลขแถวปจจุบัน บนจอภาพ
P หนาที่
ฟงกชนันี้จะท ําหนาที่ คืนตํ าแหนงของแถวที่แสดงบนจอภาพ
: ตัวอยางที่ 3.196
        USE FILEA
        DO WHILE .NOT. EOF()
           @ ++ROW(),10 SAY FIELD1
           @   ROW(),20 SAY FIELD1

           @   ROW(),30 SAY FIELD1
           SKIP
        ENDDO

&  3.164 RTRIM()
! รูปแบบ
[R]TRIM(<ขอความ>) --> ขอความที่ผานการตัดชองวางทางขวา
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ น ําชองวางทางขวาของตัวแปรตัวอักษรออกไป
: ตัวอยางที่ 3.197
        ? RTRIM("ABCD")                     // ABCD
        ? RTRIM(" A   ")+RTRIM(" B.")       //  A B.
        ? RTRIM([ A   ])+RTRIM([ B ])+"C"  //  A BC
        X = [ABC   ]
        Y = [   ABC]
        ? X+Y                               // ABC      ABC
        ? RTRIM(X)+RTRIM(Y)                 // ABC   ABC
        ? RTRIM(X)+Y                        // ABC   ABC
        ? X+RTRIM(Y)                        // ABC      ABC

&  3.165 SAVESCREEN()
! รูปแบบ
SAVESCREEN(<มมุบน>, <มมุซาย>, <มุมลาง>, <มุมขวา>) --> ตัวแปรเก็บจอภาพ
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ จัดเก็บจอภาพบางสวนไว แลวน ําออกมาแสดงอีกไดดวยคํ าสั่ง RESTSCREEN
: ตัวอยางที่ 3.198
        CLS
        SETPOS(5,10); ?? [===========================]
        SETPOS(6,10); ?? [TEST OF SAVESCREEN FUNCTION]
        SETPOS(7,10); ?? [===========================]
        ? "TEST ALL SCREEN"
        ? "---------------"
        SAVE SCREEN TO SCR1
        SCR2 = SAVESCREEN(5,10,7,36)

        CLS
        ? "THIS IS THE RESULT OF SCR1"
        RESTORE SCREEN FROM SCR1
        INKEY(0)
        CLS
        ? "THIS IS THE RESULT OF SCR2"
        RESTSCREEN(5,10,7,36,SCR2)
        RESTSCREEN(8,10,10,36,SCR2)
        RESTSCREEN(15,5,17,31,SCR2)
        RESTSCREEN(15,45,17,71,SCR2)
        INKEY(0)

&  3.166 SCROLL()
! รูปแบบ
SCROLL([<มมุบน>], [<มมุซาย>], [<มุมลาง>], [<มุมขวา>], [<จํ านวนบรรทัด>])
       --> NIL
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สั่งเลื่อนขึ้นลงแตละบรรทัด ตามขอบเขตที่ระบุ
: ตัวอยางที่ 3.199
        CLS
        SETPOS(1,5) ; ?? "ABCD"
        SETPOS(2,5) ; ?? "EFGH"
        SETPOS(3,5) ; ?? "IJKL"
        SETPOS(4,5) ; ?? "MNOP"
        SETPOS(5,5) ; ?? "QRST"
        SETPOS(6,5) ; ?? "UVWX"
        SETPOS(7,5) ; ?? "YZ.."   // แสดงขอมูลมาใหดู เห็น 7 บรรทัด
        INKEY(0)                  // ขณะนี้บรรทัดแรกอยูบรรทัดที ่1
        SCROLL(0,5,7,9,1)         // ขยับขึ้น และเห็นทั้ง 7 บรรทัด
        INKEY(0)                  // ขณะนี้บรรทัดแรกอยูบรรทัดที ่0
        SCROLL(0,5,7,9,-1)        // ขยับลง และยังเห็นทั้ง 7 บรรทัด
        INKEY(0)                  // ขณะนี้บรรทัดแรกอยูบรรทัดที ่1
        SCROLL(0,5,7,9,-1)        // ขยับลง ไมเห็นบรรทัดสุดทาย
        INKEY(0)                  // ขณะนี้บรรทัดแรกอยูบรรทัดที ่2

        SCROLL(0,5,7,9,-1)        // ขยับลง เห็นเพียง 5 บรรทัดแรก
        INKEY(0)                  // ขณะนี้บรรทัดแรกอยูบรรทัดที ่3
        CLS
        SETCURSOR(0)
        @ 0,0,24,79 BOX '.........'
        SETCOLOR("W/B")
        SETPOS( 5,10); ?? [*******************************]
        SETPOS( 6,10); ?? [--     --   --     --   -------]
        SETPOS( 7,10); ?? [ --   ---   --   ----   --     ]
        SETPOS( 8,10); ?? [  -- --     ---------   --     ]
        SETPOS( 9,10); ?? [   ---      ----   --   --     ]
        SETPOS(10,10); ?? [   ---      --     --   -------]
        SETPOS(11,10); ?? [*******************************]
        FOR I  = 5 TO 15
           SCROLL(5,10,15,40,-1)
           INKEY(0.5)
        NEXT

&  3.167 SECONDS()
! รูปแบบ
SECONDS() --> จ ํานวนวินาทีปจจุบัน
P หนาที่
ฟงกชนันี้จะท ําหนาที่ หาคาวินาทีจากเวลา ของวันนั้น
: ตัวอยางที่ 3.200
        ? TIME()                // 19:56:50
        ? SECONDS()             //  71810.41
        // ถา 20:00:00 จะได 72000 วินาที ซึ่งเกิดจาก (20 * 60 * 60)
        // ดงันัน้ 71810.41 จึงเกิดจาก
        // (19 * 3600) + (56 * 60) + 50 ได 71810
        // สวน 0.41 เปนความละเอียด ที่ไมสามารถมองเห็นใน TIME()
        // โปรแกรมขางลางนี ้แสดงการจับเวลาวา ใชเวลาเทาใดในการกดปุม
        START = SECONDS()
        INKEY(0)
        ELAPSED = SECONDS() - START

        ? ELAPSED                                 //         79.12
        ? "MINUTE : " + STR((ELAPSED / 60),2)     // MINUTE :  1
        ? "SECOND : " + STR(MOD(ELAPSED,60),5,2)  // SECOND : 19.12

&  3.168 SELECT()
! รูปแบบ
SELECT([<สมนาม>]) --> เลขที่ของพื้นที่ท ํางาน
P หนาที่
ฟงกชันนี้จะท ําหนาที่ ระบุพื้นที่ท ํางาน (WORK AREA)
: ตัวอยางที่ 3.201
        SELECT 1                // ตองสั่งเลือกพื้นที่กอนเปดแฟม
        USE FILEA
        ? SELECT()              // 1
        SELECT 2                // ตองสั่งเลือกพื้นที่กอนเปดแฟม
        USE FILEB
        ? SELECT()              // 2
        USE FILEA               // เปลี่ยนแฟมที่เปดในพื้นที ่ที ่2
        ? SELECT()              // 2
        CLEAR ALL               // ยกเลิกตัวแปร และการใชแฟมในทุกพื้นที่
        USE FILEA  NEW          // เตมิ NEW หลัง USE จะเปดพื้นที่ใหมให
        ? SELECT()              // 1
        SS = SELECT()
        USE FILEB  NEW
        ? SELECT()              // 2
        SELECT(SS)
        USE FILEA               // เปลี่ยน หรือเปดแฟมเดิมในพื้นที่ ที ่1
        ? SELECT()              // 1
        ? SELECT("FILEB")       // 2
        ? SELECT("FILEA")       // 1
        USE FILEB
        ? SELECT("FILEB")       // 2
        ? SELECT("FILEA")       // 0
        ? SELECT()              // 1

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


                                                                    บทที่ 3  :  ฟงกชัน (Function)                                                              163

&  3.169 SET()
! รูปแบบ
SET(<เลขระบตุวัเลือก>, [<คาใหม>]) --> คาที่ถูกตั้งใหม
P หนาที่
ฟงกชันนี้จะทํ าหนาที่ ปรับปรุง หรือตรวจสอบระบบ
SET VALUES DEFINED IN SET.CH

CONSTANT       VALUE TYPE      CONSTANT           VALUE TYPE
 1 _SET_EXACT          LOGICAL      20 _SET_DEVICE        CHARACTER
 2 _SET_FIXED          LOGICAL      21 _SET_EXTRA         LOGICAL
 3 _SET_DECIMALS       NUMERIC      22 _SET_EXTRAFILE     CHARACTER
 4 _SET_DATEFORMAT CHARACTER    23 _SET_PRINTER       LOGICAL
 5 _SET_EPOCH          NUMERIC      24 _SET_PRINTFILE     CHARACTER
 6 _SET_PATH           CHARACTER    25 _SET_MARGIN        NUMERIC
 7 _SET_DEFAULT       CHARACTER    26 _SET_BELL          LOGICAL
 8 _SET_EXCLUSIVE   LOGICAL      27 _SET_CONFIRM       LOGICAL
 9 _SET_SOFTSEEK       LOGICAL      28 _SET_ESCAPE        LOGICAL
10 _SET_UNIQUE        LOGICAL      29 _SET_INSERT        LOGICAL
11 _SET_DELETED      LOGICAL      30 _SET_EXIT          LOGICAL
12 _SET_CANCEL         LOGICAL      31 _SET_INTENSITY     LOGICAL
13 _SET_DEBUG          LOGICAL      32 _SET_SCOREBOARD LOGICAL
14 _SET_TYPEAHEAD  NUMERIC      33 _SET_DELIMITERS    LOGICAL
15 _SET_COLOR          CHARACTER    34 _SET_DELIMCHARS CHARACTER
16 _SET_CURSOR         NUMERIC      35 _SET_WRAP         LOGICAL
17 _SET_CONSOLE        LOGICAL      36 _SET_MESSAGE       NUMERIC
18 _SET_ALTERNATE   LOGICAL      37 _SET_MCENTER       LOGICAL
19 _SET_ALTFILE       CHARACTER    38 _SET_SCROLLBREAK  LOGICAL

: ตัวอยางที่ 3.202
        //  #INCLUDE "SET.CH"      // ไมจํ าเปนตองมีบรรทัดนี้
        ? _SET_COUNT             // มคีาเปน 38
        FOR I = 1 TO _SET_COUNT
           ? SET(I)              // พมิพคาของการเซตทั้ง 38 ตัว
           INKEY(0.5)            // ท ําใหการแสดงผลหยุด 0.5 วินาที
        NEXT
        SET FIXED ON              // ตองมีบรรทัดนี้ จึงทํ าใหตํ าแหนงทศนิยมถูกตอง
        SET(_SET_DECIMALS,5)   // คาของ _SET_DECIMALS เทากับ 3
        ? SET(3)                  //          5.00000

        ? 1                       //          1.00000
        SET(3,6)
        ? SET(3)                  //          6.000000
        ? 1                       //          1.000000

&  3.170 SETBLINK()
! รูปแบบ
SETBLINK([<ตรรกนพิจน>]) --> ผลการตั้งคาใหม
P หนาที่
ฟงกชันนี้จะท ําหนาที่ ท ําใหเกิดการกระพริบ
: ตัวอยางที่ 3.203
        SETCOLOR("*W/B,*B/W")           // ใชคูกับ SETBLINK
        A = SPACE(3)
        SETBLINK(.F.)
        @ 10,10 SAY "ABCDE" GET A
        READ
        SETBLINK(.T.)                   // ในเซตสีตองม ี* จึงทํ าใหกระพริบ
        @ 11,10 SAY "ABCDE" GET A
        READ
        SETBLINK(.F.)
        @ 12,10 SAY "ABCDE" GET A
        READ

&  3.171 SETCANCEL()
! รูปแบบ
SETCANCEL([<ตรรกนพิจน>]) --> ผลการตั้งคาใหม
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ระบุไมใหหยุดโปรแกรม แบบทันทีทันใดดวยปุม CTRL-BREAK หรือ ALT-C
: ตัวอยางที่ 3.204
        SETCANCEL(.F.)
        FOR I = 1 TO 10       // แตโปรแกรมจะหลุด เมื่อกดปุมใดๆ 10 ครั้ง
            X = INKEY(0)
            IF X = 1          // กดปุม HOME เพื่ออนุญาติใหหยุดโปรแกรมได
               SETCANCEL(.T.) // โดยหยุดโปรแกรมเมื่อกดปุม

            ENDIF             // ALT-C หรือ CTRL-BREAK
        NEXT

&  3.172 SETCOLOR()
! รูปแบบ
SETCOLOR([<ขอความแสดงส>ี]) --> ขอความแสดงสี
P หนาที่
ฟงกชันนี้จะทํ าหนาที่ ระบุสีใหจอภาพ
: ตัวอยางที่ 3.205
        SETCOLOR("W/N,BG+/B,B,,B/W") // STANDARD   W/N
        @ 5,5 TO 8,75 DOUBLE         // ENHANCED   BG+/B ส ําหรับ GET
        X = SPACE(5)                 // BORDER     B เสนขอบนอกสุด
        Y = SPACE(5)                 // BACKGROUND (NO SUPPORTED)
        @ 6,10 SAY "TEST 1 :" GET X  // UNSLECTED  B/W  GET ทีย่ังไมถูกเลือก
        @ 7,10 SAY "TEST 2 :" GET Y
        READ
        FORM1 = "B/W,B/BG+,W,,GR+/W"
        SETCOLOR(FORM1)
        @ 10,10 SAY "TEST 3 :" GET Y
        READ

&  3.173 SETCURSOR()
! รูปแบบ
SETCURSOR([<เลขระบุแบบตัวกระพริบ>]) --> เลขระบุแบบตัวกระพริบ
P หนาที่
ฟงกชันนี้จะท ําหนาที่ เปลี่ยนรูปแบบของตัวกระพริบ (CURSOR)
: ตัวอยางที่ 3.206
        SETCURSOR(0)            // NONE CURSOR
        INKEY(0)
        SETCURSOR(1)            // UNDERLINE CURSOR & SET CURSOR ON
        INKEY(0)
        SETCURSOR(2)            // LOWER HALF-BLOCK CURSOR
        INKEY(0)  ;        SETCURSOR(3)            // FULL BLOCK CURSOR
        INKEY(0)  ;        SETCURSOR(4)            // UPPER HALF-BLOCK CURSOR

        INKEY(0)  ;        SET CURSOR OFF          // NONE CURSOR
        INKEY(0)

&  3.174 SETKEY()
! รูปแบบ
SETKEY(<เลขปุมบรรแปนพิมพ>, [<กลองปฏิบัติการ>]) --> กลองปฏิบัติการ
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ มอบหมายการกดปุมใหไปทํ าโปรแกรมยอยที่ตองการ
: ตัวอยางที่ 3.207
        #DEFINE HOME 1
        SETKEY(HOME,{|X|H()})           // HOME
        SETKEY(27  ,{|X|E()})           // ESC
        A = INKEY(0)
        Z = SETKEY(A)
        IF Z != NIL ; EVAL(Z) ; ENDIF
        FUNCTION H
           ? "YOU PRESS HOME"
        RETURN
        FUNCTION E
           ? "YOU PRESS ESC"
        RETURN

&  3.175 SETMODE()
! รูปแบบ
SETMODE(<จ ํานวนแถว>, <จํ านวนหลัก>) --> ผลการตั้งคาใหม
P หนาที่
ฟงกชันนี้จะท ําหนาที่ ตรวจสอบ MODE การแสดงผล วาใชกับ MODE ใดได
: ตัวอยางที่ 3.208
        ? SETMODE(25,80)        // .T.
        ? SETMODE(43,80)       // .T.
        ? SETMODE(10,50)        // .F.
        IF SETMODE(25,80)
           ? "25 ROWS AND 80 COLUMNS AVAILABLE"
        ELSE

           ? "INVALID MODE"
        ENDIF

&  3.176 SETPOS()
! รูปแบบ
SETPOS(<เลขแถว>, <เลขหลัก>) --> NIL
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ยายตํ าแหนงของตัวกระพริบ(CURSOR) บนจอภาพ
: ตัวอยางที่ 3.209
        SETPOS(5,0)
        ?? "---"
        SETPOS(5,4)
        ?? "---"                // --- ---
        ?  "---"                // ---
        FOR I = 1 TO 3                          //  12
          FOR J = 1 TO 2                        //  12
             SETPOS(7+I,J) ; ?? LTRIM(STR(J))  //  12
          NEXT
        NEXT
        FOR I = 1 TO 3                          //  11
          FOR J = 1 TO 2                        //  22
             SETPOS(10+I,J) ; ?? STR(I,1,1)    //  33
          NEXT
        NEXT

&  3.177 SETPRC()
! รูปแบบ
SETPRC(<เลขแถว>, <เลขหลัก>) --> NIL
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ เซตตํ าแหนงของหัวพิมพบนเครื่องพิมพ เพื่อเปลี่ยนคา PROW() และ PCOL()

: ตัวอยางที่ 3.210
        SETPRC(1,1)
        SETPRC(PROW()+1,PCOL())
        SETPRC(PROW(),PCOL()+1)
        SETPRC(PROW()+1,10)

&  3.178 SPACE()
! รูปแบบ
SPACE(<จ ํานวนเลข>) --> ขอความที่เปนชองวาง
P หนาที่
ฟงกชนันี้จะท ําหนาที่ เปลี่ยนตัวอักษรเปนชองวาง
: ตัวอยางที่ 3.211
        X = SPACE(5)
        Y = SPACE(10)
        Z = 20
        A = SPACE(Z)
        FILEA->FIELD1 = SPACE(LEN(FILEB->FIELD1))
        INKEY(0)
        FILEA->FIELD1 = LASTKEY()

&  3.179 SQRT()
! รูปแบบ
SQRT(<จ ํานวนเลข>) --> ผลการถอดรากที่ 2
P หนาที่
ฟงกชันนี้จะท ําหนาที่ สงคารากที่สอง
: ตัวอยางที่ 3.212
        SET DECIMALS TO 7
        ? SQRT(4)               // 2.0000000
        SET DECIMALS TO 5
        ? SQRT(2)               // 1.41421
        SET DECIMALS TO 3
        ? SQRT(9)               // 3.000
        ? SQRT(9) ** 2          // 9.000
        ? SQRT(9) ** 0          // 1.000
        ? SQRT(9) ** 1          // 3.000

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


                                                                    บทที่ 3  :  ฟงกชัน (Function)                                                              169

        ? SQRT(4) ** 3          // 8.000
        ? SQRT(4) ^ 3           // 8.000
        ? SQRT(81) ^ 2          // 81.000
        X = 5 * 5
        ? SQRT(X)               // 5.000

&  3.180 STR()
! รูปแบบ
STR(<ตวัเลข>, <ความยาว>, <จํ านวนทศนิยม>) --> ขอความ
P หนาที่
ฟงกชันนี้จะท ําหนาที่ เปลี่ยนตัวเลขเปนตัวอักษร
: ตัวอยางที่ 3.213
        ? "*"                   // *
        ? STR(5,6,2)            //   5.00
        ? STR(5.25,6,2)         //   5.25
        ? STR(5.25,7,2)         //    5.25
        ? STR(5.25,8,2)         //     5.25
        ? STR(5.25,1,2)         // 5
        ? STR(5.25,2,2)         // **   แสดงทศนิยมไมได
        ? STR(5.25,3,2)         // ***  แสดงทศนิยมไมได
        ? STR(5.25,4,2)         // 5.25
        ? STR(5.25,4)           //    5
        ? STR(12.69,1,2)        // *    ขนาดเล็กเกินไปที่จะแสดงตัวเลข
        ? STR(12.69,2,2)        // **
        ? STR(12.69,3,2)        // ***
        ? STR(12.69,3)          //  13
        ? STR(12.69,5,2)        // 12.69
        ? STR(12.69,4,1)        // 12.7
        X = 5
        ? STR(X,3)          //   5
        X = 5.2499
        ? STR(X,3,1)            // 5.2
        ? STR(X,3)              // 5

&  3.181 STRTRAN()
! รูปแบบ
STRTRAN(<ขอความหลัก>, <ขอความที่นํ าไปคนหา>,
        [<ขอความทีน่ ําไปแทนที>่], [<ตํ าแหนงเริ่ม>], [<จํ านวนตํ าแหนง>])
        --> ขอความใหม
P หนาที่
ฟงกชนันี้จะท ําหนาที่ คนหาและแทนที่ขอความดวยขอความ
: ตัวอยางที่ 3.214
        ? STRTRAN("ABCDEFGH","C","C")   // ABCDEFGH
        ? STRTRAN("ABABABAB","B","A")  // AAAAAAAA
        X = UPPER(STRTRAN("ABCDEFGH","C","C"))
        ? X
        X = X + X
        ? STRTRAN(X,"ABC","123")        // 123DEFGH123DEFGH

&  3.182 STUFF()
! รูปแบบ
STUFF(<ขอความหลกั>, <ตํ าแหนงเริ่ม>, <จํ านวนลบ>, <ตัวอักษรที่แทรกเขาไป>)
      --> ขอความใหม
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ ลบและแทรกขอความ ดวยขอความที่ตองการในคํ าสั่งเดียวกัน
: ตัวอยางที่ 3.215
        ? STUFF("ABCDEF",3,1,"")        // ABDEF
        ? STUFF("ABCDEF",3,1,"C")       // ABCDEF
        ? STUFF("ABCDEF",3,2,"C")       // ABCEF
        ? STUFF("ABCDEF",4,3,"CBA")     // ABCCBA
        X = "MANY PEOPLE AT BANGKOK"
        ? STUFF(X,1,15,"")              // BANGKOK
        ? STUFF(X,1,0,"HI! ")           // HI! MANY PEOPLE AT BANGKOK
        ? STUFF(X,13,2,"IN")            // MANY PEOPLE IN BANGKOK

&  3.183 SUBSTR()
! รูปแบบ
SUBSTR(<ขอความหลกั>, <ตํ าแหนงเริ่ม>, [<จํ านวนตํ าแหนง>]) --> ขอความใหม
P หนาที่
ฟงกชนันีจ้ะท ําหนาที่ แยกขอความที่ตองการจากขอความ
: ตัวอยางที่ 3.216
        X = "LAMPANG , THAILAND"
        X1 = SUBSTR(X,1,7)
        X2 = SUBSTR(X,11,8)
        Y = X1 + " , " + X2
        ? X                             // LAMPANG , THAILAND
        ? Y                             // LAMPANG , THAILAND

&  3.184 TIME()
! รูปแบบ
TIME() --> ขอความเวลา
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แสดงเวลาของระบบ
: ตัวอยางที่ 3.217
        ? TIME()                        // 13:52:29
        X = TIME()
        ? SUBSTR(X,1,2)                 // 13
        ? SUBSTR(X,4,2)                 // 52
        ? SUBSTR(X,7,2)                 // 29
        A1 = VAL(SUBSTR(TIME(),1,2))
        A2 = SUBSTR(TIME(),4,2)  ;      A3 = TIME()
        A = A1 + VAL(A2) + VAL(SUBSTR(A3,7,2))
        ? A                             // 94

&  3.185 TONE()
! รูปแบบ
TONE(<ความถี>่, <ความนาน>) --> NIL
P หนาที่
ฟงกชันนี้จะท ําหนาที่ ฟงกชันออกเสียงผานลํ าโพง

โน็ต และความถี่
C   130.80       G  196.00    MID C  261.70      G  392.00
C# 138.60      G# 207.70        C# 277.20      G# 415.30
D  146.80       A  200.00        D  293.70      A  440.00
D# 156.60       A# 233.10        D# 311.10     A# 466.20
E  164.80       B  246.90        E  329.60      B  493.90
F  174.60                  F  349.20      C  523.30
F# 185.00                 F# 370.00

: ตัวอยางที่ 3.218
        TONE(130.80,18)         // เสียงโน็ต C นาน 1 วินาที
        TONE(146.80,36)         // เสียงโน็ต D นาน 2 วินาที
        TONE(130.80,1)           // เสียงโน็ต C นาน 1/18 วินาที
        TONE(130.80,48)         // เสียงโน็ต C นาน 3 วินาที

&  3.186 TRANSFORM()
! รูปแบบ
TRANSFORM(<นพิจน>, <รูปแบบที่ตองการ>) --> นิพจนใหม
P หนาที่
ฟงกชนันี้จะท ําหนาที่ แปลงคาใหไปอยูในรูปแบบของตัวอักษร

ตวัแบบที่ใชเปลี่ยนรูปแบบ
A,N,X,9,#   ส ําหรับการแปลงอักษร และตัวเลข
L           แสดงคา T,F ส ําหรับคาแบบตรรกะ
Y           แสดงคา Y,N ส ําหรับคาแบบตรรกะ
!           แสดงตัวอักษรตัวใหญ
$,*         แสดงเครื่องหมายนี้หนาตัวเลข
. หรือ ,    แตงขอมูลตัวเลขใหเขาใจไดงาย
ฟงกชันที่ใชเปลี่ยนรูปแบบ
B           ใหขอมูลแสดงแบบชิดซาย และมักใชกับตัวเลข
C           เติม CR หลังคาที่เปน +
X           เติม DR หลังคาที่เปน -
D           แสดงวันที่ตามรูปแบบที่ก ําหนดดวย SET DATE
E           แสดงวนัที่ตามแบบของอังกฤษ (BRITISH)
R           ไมมีการจัดรูปแบบ
Z           ถาม ี0 หนาตัวเลขใหแสดงชองวางแทน
(           คาลบ ใหอยูภายในเครื่องหมาย ()
!           ใหเปลี่ยนเปนตัวอักษรใหญ

: ตัวอยางที่ 3.219
        ? TRANSFORM(123,"@C")           //        123 CR
        ? TRANSFORM(-123,"@X")          //        123 DB
        ? TRANSFORM(-123,"@R")          //       -123
        ? TRANSFORM(-123,"")            //       -123
        ? TRANSFORM(-123,"C")           // C
        ? TRANSFORM(-123,"@B")          // -123
        ? TRANSFORM(-0123,"@B")         // -123
        ? TRANSFORM(-01230,"@Z")        //      -1230
        ? TRANSFORM(1234.56,"9999999")  //    1235
        ? TRANSFORM(1234.56,"$99,999")  // $ 1,235
        ? TRANSFORM("ABCD","@!")        // ABCD
        ? TRANSFORM("ABCD","!AA")       // ABC
        SET DATE TO ANSI                // YY.MM.DD
        ? TRANSFORM(DATE(),"@D")        // 96.03.13
        SET DATE TO AMERICAN
        ? TRANSFORM("04/15/96","@D")    // 04/15/96  ตามที ่SET DATE
        ? TRANSFORM("04/15/96","@E")    // 15/04/96  BRITISH
        ? TRANSFORM(123,"$9999")        // $ 123
        ? TRANSFORM(-123,"@(")          // (      123)
        ? TRANSFORM("-123","@(")        // -123

&  3.187 TYPE()
! รูปแบบ
TYPE(<นพิจน>) --> แบบของนิพจน
P หนาที่
ฟงกชันนี้จะท ําหนาที่ ระบุแบบของขอมูลในชุดเงื่อนไขของนิพจน

แบบของนิพจน
A  ARRAY                L  LOGICAL        U  NIL,LOCAL OR STATIC
B  BLOCK                M  MEMO          UE ERROR SYNTACTICAL
C  CHARACTER    N  NUMERIC        UI ERROR INDETERMINATE
D  DATE                 O  OBJECT

: ตัวอยางที่ 3.220
        ? TYPE("LEFT('12AA',2)")        // C
        ? TYPE("VAL(LEFT('12AA',2))")   // N
        ? TYPE("IF(5>2,'MORE',10)")     // C
        ? TYPE("IF(5<2,'MORE',10)")     // N
        ? TYPE("A")                     // U
        ? TYPE('B')                     // U
        ? TYPE("C()")                   // U
        ? TYPE("UDF()")                 // U
        ? TYPE('IF(2-3)')               // UE
        ? TYPE('.T.')                   // L
        ? TYPE("IF(2<3,.T.,.F.)")       // L

&  3.188 UPDATED()
! รูปแบบ
UPDATED() --> ตรรกแจงวามีการปรับปรุงขอมูลหรือไม
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ตรวจสอบการปรับปรุงขอมูล จากผลการ GET ... READ
: ตัวอยางที่ 3.221
        A = 5
        B = "TEST"
        @ 5,5 SAY "GET A : " GET A
        @ 6,5 SAY "GET B : " GET B
        READ
        IF UPDATED()
           ? "YOU CAN MANAGE WITH YOUR FILES."
        ELSE
           ? "NO UPDATED"
        ENDIF

&  3.189 UPPER()
! รูปแบบ
UPPER(<ขอความ>) --> ขอความที่เปนตัวใหญหมด
P หนาที่
ฟงกชนันี้จะท ําหนาที่ เปลี่ยนตัวอักษรใหเปนตัวใหญหมด
: ตัวอยางที่ 3.222
        ? UPPER("AbC")          // ABC
        ? UPPER("abC")          // ABC

&  3.190 USED()
! รูปแบบ
USED() --> ตรรกแจงวามีการเปดแฟมหรือไม
P หนาที่
ฟงกชนันี้จะท ําหนาที่ ตรวจสอบวามีการใชแฟมอยูหรือไม
: ตัวอยางที่ 3.223
        USE FILEA
        ? USED()                // .T.
        SELE 2
        ? USED()               // .F.
        SELE 1
        ? USED()                // .T.
        CLOSE
        ? USED()                // .F.

&  3.191 VAL()
! รูปแบบ
VAL(<ขอความที่เปนตัวเลข>) --> ตัวเลข
P หนาที่
ฟงกชันนี้จะท ําหนาที่ เปลี่ยนตัวอักษรเปนตัวเลข
: ตัวอยางที่ 3.224
        ? VAL("1.94499")                // 1.94499
        ? VAL("1.945")                  // 1.945
        ? VAL(" 1.945")                 //  1.945
        ? VAL(" 1.945 ")                //  1.9450
        ? VAL(" 1. 945 ")               //  1.00000
        ? VAL(" 5. ") + VAL(" 11 ")     //       16.0
        ? VAL("A")                      //   0
        ? VAL(" A")                     //   0
        ? VAL(SPACE(0))                 //   0
        ? VAL(SPACE(1))                 //   0
        ? VAL(SPACE(3))                 //   0

&  3.192 VALTYPE()
! รูปแบบ
VALTYPE(<นพิจน>) --> แบบของนิพจน

P หนาที่
ฟงกชันนี้จะท ําหนาที่ ระบุแบบขอมูลของนิพจน

แบบที่ใชกํ าหนดนิพจน
A  ARRAY                L  LOGICAL        U  NIL
B  BLOCK                M  MEMO
C  CHARACTER            N  NUMERIC
D  DATE                 O  OBJECT

: ตัวอยางที่ 3.225
        X = ARRAY(20)
        ? VALTYPE("UNIVERSITY")         // C
        ? VALTYPE(22/7)                 // N
        ? VALTYPE(X)                    // A
        ? VALTYPE(CTOD("12-31-96"))     // D
        ? VALTYPE(.T.)                  // L
        ? VALTYPE(3 > 5)                // L
        ? VALTYPE("")                   // C
        ? VALTYPE(NIL)                  // U

&  3.193 VERSION()
! รูปแบบ
VERSION() --> ชือ่เวอรชั่นของภาษาที่ใช
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แสดงเลข VERSION ของ CLIPPER
: ตัวอยางที่ 3.226
        ? VERSION()             // CLIPPER (R) 5.01
        ? OS()                  // DOS 6.20

&  3.194 YEAR()
! รูปแบบ
YEAR(<วนัที>่) --> ป ค.ศ.
P หนาที่
ฟงกชันนี้จะท ําหนาที่ แปลงวันที่เปนเลขปที่ตองการ
: ตัวอยางที่ 3.227
        ? DATE()                       // 12/31/02
        ? YEAR(DATE())                 // 2002
        ? YEAR(CTOD("05/11/95"))      // 1995
        ? YEAR(DATE()+5)               // 2003

